Lab 5 - Outputs - Writing to Ports

Instructor: Fred Etcheverry

Review, Chapter 7, Parallel I/O Ports, ***MC 68HC11 An Introduction Software and Hardware Interfacing*** by Han-Way Huang. ***CME11E9-EVBU Development Board Instruction*** Reference: ***Manual***. Download form www.axman.com.

Equipment: PC computer, ***CME-11E9-EVBU Development Board***. Purchased from Axiom Manufacturing Inc. www.axman.com.

Parts: 1-TIP120 NPN Darlington Transistor, 1- 100Ω and 1-1KΩ 1/4W resistors, 1-8Ω speaker, and #22 solid hook-up wire.

Software: AxIDE v. 2. comes with Development Board or can be downloaded.

Constructing Interface

The ***CME-11E9-EVBU Development Board*** provides a ***Prototype Board*** on which to construct interface circuits. This board consists of an array of small sockets that make connections to #22 gage solid wire (American Wire Gage, OD = 0.6438 millimeters) or component leads. The row sockets at the top and bottom of the board are connected horizontally on the back of the board such that any wire or lead that is inserted into a row socket is connected to every other socket in that row. Usually, one set of row sockets are

Fig. 5.1 Prototyping Board



connected to Vcc, while the other is connected to ground (Fig. 5.1). The column sockets are connected vertically. A notch separates the columns and facilitates DIPs (Dual in-line Pins — electronic packages with two rows of leads.) We will use DIPs in future labs.

1. **Connect** the following circuit to the Development Board.

Fig. 5.2 Pictorial and Schematic Diagram of Audio Interface.





The output of PD4 feeds the base of Darlington transistor TIP120 through a 1K-ohm resistor.

The Darlington amplifies the signal from the port to drive the speaker. The 100-ohm resistor is provided to limit the current drawn from the Development Board's supply. If a separate supply is available this resistor can be eliminated and the sound will be much louder. Caution: Connect the supply's negative terminal to the Development Board's ground but do not connect the supply's positive terminal to the Development Board's supply. Connect the positive terminal only to the speaker. A supply voltage greater than 5-volts may be used to produce more sound power, but a larger speaker should be used. The TIP120 can handle much more power than is required in this lab especially if it provided with a heat sink.

While the TIP120 is shown in the schematic as a regular bipolar transistor, the actual drawing of a darlington is shown in the upper right corner. The pin-out of the TIP120 is shown in the lower right corner.

Port Driver

2. Assemble, list, and run the following program.

```
*This program will produce a signal out Port D.
        ORG     $0200
PORTD   EQU     $1008   ;ADDRESS OF PORT D
DDRD    EQU     $1009   ;ADDRESS OF PORT D CONTROL REGISTER
START   LDAA    #$3F    ;VALUE TO LOAD IN DDRD TO
        STAA    DDRD    ;CONFIGURE PORT D TO OUTPUT.
        LDAA    #$FF    ;GET ALL ONES IN A.
AGAIN   LDAB    #$FF    ;GET COUNTER IN B.
        COMA            ;TOGGLE CONTENTS OF A TO MAKE WAVE.
LOOP    DECB            ;NOT DONE THEN
        BNE     LOOP    ;KEEP LOOPING, ELSE,
        STAA    PORTD   ;OUTPUT PORT B AND GO BACK
        BRA     AGAIN   ;AND TOGGLE WAVE.
        END
```

When this program is run the speaker should emit a constant tone.

3. Analyze the program. Which constant controls the frequency of the tone? _____________.

4. Use the MM instruction to change this constant to make the tone higher.

5. Use the MM instruction to change this constant to make the tone lower.

Calculating Frequency

6. Calculate the frequency of the tone. ________

To do this you must find the time it takes the program to output one complete cycle from Port D. This depends on the speed of the processor and the number of instructions that must be executed. The 68HC11 on the Development Board is clocked by an 8MHz crystal oscillator. This clock and Buffalo set the instruction cycle frequency to 2MHz. This will be explained in future labs. There are thus 2-million instruction cycles a second. The period of each cycle is thus 1/(2-million) = 1/2-microseconds (millionth of a second).

Calculate the number of instruction cycles to complete one complete output cycle from Port D. To find the number of instruction cycles required by each instruction, look at the listing or check the particular instruction in Appendix D, p(621). Let’s look at the listing of the above program. Notice the bracketed numbers in the middle of this listing. They show the number of instruction cycles of each instruction.

Fig. 5.3 Nested Loops Generate Tone



The above program segment (Fig. 5.3) taken from the above listing produces an output waveform that drives the speaker to create a tone. Each time this segment executes, Register B is loaded with $FF (decimal 256). The inner loop produces a delay by counting down on Register B until its content is zero. The STAA instruction outputs the contents of Register A. If the content is all ones, the output is high (about 5-volts). If the contents are all zeros, the output is low (about zero volts or ground). Each time this segment repeats the COMA instruction compliments the content of Register A — zeros become ones and ones becomes zeros. This complimenting causes the output to toggle as each consecutive execution of this segment causes the output to change from high to low. It thus takes two executions of this segment to produce a complete waveform (Fig. 5.4).

By calculating the execution time of this segment and doubling it, we can find the ***time of the period*** of the waveform produced. Dividing the time of the period into one will give the frequency. (Fig. 5.5) The inner loop (Fig. 5.4) require 256 x (3cyc. + 2cyc.) = 1280cyc. Each execution of the outer loop requires 2cyc. + 2cyc. + 1280cyc. + 4cyc. + 3cyc. = 1291cyc. This is the number of cycles necessary to complete a half of a waveform. To complete a whole waveform thus require 2 x 1291cyc. = 2582cyc. Since each cycle takes 0.5-microseconds, the total time to complete a waveform (time of the period) is thus 2582cyc. x 0.5-microseconds/cyc = 1291-microseconds. Frequency is one divided by the time of period = 1/1291-microseconds or about 775Hz.

Fig. 5.4 Waveform timing



$$T_p = (2582\text{cyc.}) \times (0.5\text{uS/cyc.}) = 1291\text{uS}$$

Where Tp is the time of the period.

$$F = \frac{1}{T_p} = \frac{1}{1291\text{uS}} = 774.5\text{Hz} \approx 775\text{Hz}$$

Where F is the frequency of the waveform.

7. Modify the above program to output of 1KHz signal (about).

8. Assemble and run the following program.

```
*This program will produce a signal out Port D that will go from
*low to high.
        ORG     $0200
PORTD   EQU     $1008   ;ADDRESS OF PORT D
DDRD    EQU     $1009   ;ADDRESS OF PORT D CONTROL REGISTER
START   LDAA    #$3F    ;VALUE TO LOAD IN DDRD TO
        STAA    DDRD    ;CONFIGURE PORT D TO OUTPUT.
        LDAA    #$FF    ;GET ALL ONES IN A.
        LDAB    #$01    ;GET COUNTER IN B.
AGAIN   COMA            ;TOGGLE CONTENTS OF A TO MAKE WAVE.
        BSR     DELAY   ;VARIABLE DELAY
        STAA    PORTD   ;OUTPUT PORT B AND GO BACK
        BRA     AGAIN   ;AND TOGGLE WAVE.
DELAY   PSHA            ;SAVE A.
        DECB            ;
        BNE     AHEAD
        DEC     $0230
        LDAB    #$10
AHEAD   LDAA    $0230   ;GET VARIABLE.
LOOP1   DECA            ;WASTE
        BNE     LOOP1   ;TIME.
        PULA            ;RESTORE A
        RTS
        END
```

9. Find the address of the constant that determines how fast the tone rises.
Address = ______________

10. Use the MM instruction to slow down the rate that the tone rises.

11. Find the instruction that causes the tone to rise and modify it to make the tone fall. Instruction = ______________________

As you might suspect sounds rising or falling in pitch attract attention. The concepts in this lab can be used to make an alarm.